

Solución de ecuaciones diferenciales de segundo orden con series de potencia.

Ejemplo 1

Tarea 1. Construya un programa que permita hallar la solución de una ecuación diferencial de segundo orden por medio de una serie de potencias en la vecindad del punto ordinario $x = 0$.

a. Aplique el programa para resolver la ecuación de Airy $y'' - xy = 0$

Solución

El recuadro muestra el programa "SerieP" que permite hallar los coeficientes de la serie parcial $y = \sum_{n=0}^N a_n x^n$, con $N > 3$. Para usar el programa, la ecuación debe estar escrita en la

forma: $y'' = P(x)y' + Q(x)y$

```

DelVar P,Q,N,T,E,E1,E2,h,x,f,A
Input P,"P"
Input Q,"Q"
Input N,"N>3"
((P|x=0)×a1+(Q|x=0)×a0)/2⇒A
{a0,a1,A}⇒T
(P×diff(h(x),x)+Q×h(x))⇒E
For 1⇒k To N-3
simplify(diff(E,x))⇒E
simplify(E)|diff(h(x),x)=a1⇒E1
For 2⇒j To k+1
simplify(E1)|diff(h(x),x,j)=j!×T[j+1]⇒E1
Next
simplify(E1)|h(x)=a0|x=0⇒E2
E2/((k+2)!)⇒f
augment(T,{f})⇒T
Next
PrintNatural T

```

El usuario deberá introducir las funciones $P(x)$ y $Q(x)$, además del número N de términos a considerar en la serie parcial.

a. Para resolver la ecuación de Airy, escribimos $y'' = xy$, esto implica que $P(x) = 0$ y $Q(x) = x$. El resultado para $N = 7$ se muestra en la figura 1.

La serie puede escribirse en la forma:

$$y = a_0 + a_1 x + \frac{1}{6} a_0 x^3 + \frac{1}{12} a_1 x^4 + \frac{1}{180} a_0 x^6 + \dots$$

O de otra manera:

$$y = a_0 \left(1 + \frac{1}{6} x^3 + \frac{1}{180} x^6 + \dots \right) + a_1 \left(x + \frac{1}{12} x^4 + \dots \right)$$




Figura 1

Tarea 2. Construya un programa que permita obtener y graficar las soluciones particulares, es decir dados los valores específicos para los coeficientes a_0 y a_1 , de la ecuación diferencial $y'' = P(x)y' + Q(x)y$ por medio de series de potencias en la vecindad del punto ordinario $x = 0$.

- a. Con el propósito de estudiar gráficamente la convergencia de la serie obtenida considere la ecuación diferencial $y'' + y = 0$, con las condiciones iniciales $y(0) = 0$, $y'(0) = 1$. Obtenga su solución mediante serie de potencia con el programa confeccionado y grafique para valores crecientes de N . Explique los resultados obtenidos.

Solución.

El programa “SerieGra” se muestra en el recuadro. Este programa utiliza como subrutina el programa “Serie2” que se muestra en la figura 2.

```

DelVar P,Q,a,b,T,T1,sp,B,Z,D,U
Input P, "P"
Input Q, "Q"
Input N, "N>3"
Input a, "a0"
Input b, "a1"
Serie2(P,Q,N)
T|a0=a|a1=b => T1
a => sp
For 2 => k To N
sp+T1[k]×x^(k-1) => sp
Next
ClrGraph
ClearSheet 1
SetSimulGraph On
DefaultSetup
ClrGraph
Input B, "Xmin", "ViewWindow"
Input Z, "Xmax", "ViewWindow"
Input D, "Ymin", "ViewWindow"
Input U, "Ymax", "ViewWindow"
Wait 1
ViewWindow B,Z,1,D,U,1
GTSELon 1
GraphType "y="
Print "Suma Parcial:"
Print sp
Define y1(x)=sp
DrawGraph
DelVar P,Q,a,b,T,T1,B,Z,D,U
  
```

Observe que este último programa usa los parámetros “P, Q, N” y devuelve una lista “T” con N coeficientes de la suma parcial y que serán empleados por el programa “SerieGra” para construir un polinomio y graficarlo. Para aplicar el programa la ecuación diferencial debe escribirse en la forma




Figura 2

$y'' = P(x)y' + Q(x)y$. Las condiciones iniciales determinan los dos primeros coeficientes. Tenemos que $y(0) = a_0$ y $y'(0) = a_1$.

Conocemos, por otros métodos que la solución general de la ecuación $y'' + y = 0$ está dada por:

$$y = A \cos x + B \sin x .$$


Para obtener la solución particular que satisfaga las condiciones iniciales dadas debemos escribir:

$$y(0) = A \cos 0 + B \sin 0 = A = 0$$

$$y'(0) = B \cos 0 = B = 1 .$$

Por tanto la solución particular es $y = \sin x$.

En general no siempre es posible hallar las soluciones de ecuaciones diferenciales con ayuda de funciones elementales, es en estos casos donde el método de las series de potencia es útil. Vamos a utilizar la ecuación $y'' + y = 0$ con las condiciones iniciales $y(0) = 0$, $y'(0) = 1$, cuya solución parcial conocemos para estudiar la convergencia de la serie obtenida mediante el programa “SerieGra”.



Datos que deben ingresarse al programa “SerieGra”:

- “P” \Rightarrow 0
- “Q” \Rightarrow -1
- “N>3” \Rightarrow 5
- “a0” \Rightarrow 0
- “a1” \Rightarrow 1
- “Xmin” \Rightarrow -9.5,
- “Xmax” \Rightarrow 9.5,
- “Ymin” \Rightarrow -1.6,
- “Ymax” \Rightarrow 1.6

El resultado obtenido se muestra en la figura 3

Figura 3

Repetiendo este procedimiento para los valores $N=7$ y $N=9$ se observa una convergencia de la suma parcial a la función $y = \sin x$, que es la solución particular de la ecuación diferencial $y'' + y = 0$, con las condiciones iniciales $y(0) = 0$, $y'(0) = 1$. (Ver figuras 4-6)




Figura 4




Figura 5




Figura 6

Al aplicar el método de series de potencia en la vecindad de un punto ordinario, estamos encontrando la serie de Taylor de la solución desarrollada en la vecindad de dicho punto. La figura 7, muestra el resultado parcial para $N = 9$ y la correspondiente serie de Taylor para la función $y = \sin x$ en la vecindad de $x=0$. Note que el resultado, al aplicar el programa, queda guardado en la variable “sp” que podemos manipular en futuros cálculos o gráficos.




Figura 7

Tarea 3. Construya un programa para encontrar la ecuación indicial y resolverla, en el caso de soluciones mediante series de potencia en la vecindad de un punto singular regular de ecuaciones en la forma:

$$P(x)y'' + Q(x)y' + R(x)y = 0$$

- a. Utilice el programa para calcular las raíces de la ecuación indicial de la ecuación diferencial $2x^2y'' - xy' + (1+x)y = 0$

Solución.

Vamos a suponer que $x=0$ es un punto singular regular de la ecuación $P(x)y'' + Q(x)y' + R(x)y = 0$. Dividimos la ecuación por $P(x)$ y multiplicamos por x^2 . Obtenemos:

$$x^2y'' + x[xp(x)]y' + [x^2q(x)]y = 0, \text{ donde } p(x) = \frac{Q(x)}{P(x)} \text{ y } q(x) = \frac{R(x)}{P(x)}$$

La ecuación indicial se determina mediante la relación $F(r) = r(r-1) + p_0r + q_0 = 0$. Donde $p_0 = \lim_{x \rightarrow 0} x \cdot p(x)$ y $q_0 = \lim_{x \rightarrow 0} x^2q(x)$. El programa "Indicial", que usa estas formulas se muestra en el recuadro.

- a. Las raíces de la ecuación indicial para la ecuación $2x^2y'' - xy' + (1+x)y = 0$ se muestran en la figura 8

```

DelVar P,Q,R,p0,r,q0,S
Input P, "P"
Input Q, "Q"
Input R, "R"
lim(x*Q/P,x,0) => p0
lim(x^2*R/P,x,0) => q0
solve(r*(r-1)+p0*r+q0=0,r) => S
PrintNatural S, "Raices"
DelVar P,Q,R,p0,r,q0
    
```




Figura 8

Tarea 4. Construya un programa para hallar “N” de los coeficientes de la serie de potencias en la vecindad de un punto singular regular de una ecuación de la forma

$P(x)y'' + Q(x)y' + R(x)y = 0$ si se conocen las raíces r_1 y r_2 de la ecuación indicial. (Método de Frobenius)

a. Utilice el programa con “N = 5” términos de la solución general de la ecuación:

$$2x^2 y'' - xy' + (1 + x^2)y = 0.$$

```

DelVar P,Q,R,N,p,q,ra1,T
Input P, "P"
Input Q, "Q"
Input R, "R"
Input N, "Número de términos"
Input ra1, "Raiz Ec. Indicial"
taylor(x×Q/P,x,N)⇒p
taylor(x^2×R/P,x,N)⇒q
CoefPx(p,N)
T⇒Tp
DelVar T
CoefPx(q,N)
T⇒Tq
DelVar T
x×(x-1)+Tp[1]×x+Tq[1]⇒FI
{a0}⇒T
For 1⇒n To N
0⇒su
For 0⇒k To n-1
su+T[k+1]×((ra1+k)×Tp[n-k+1]+Tq[n-k+1])⇒su
Next
-su/(FI|x=(ra1+n))⇒g
augment(T,{g})⇒T
Next
a0⇒res
For 2⇒j To N
res+T[j]×x^(j-1)⇒res
Next
PrintNatural x^(ra1)×res,"Serie de Potencias"

```

Solución

En el recuadro se muestra el programa “Frobe” que permite calcular la solución de una ecuación diferencial de la forma $P(x)y'' + Q(x)y' + R(x)y = 0$ mediante una serie de potencias en la vecindad de un punto singular regular (Método de Frobenius). Note que el programa debe aplicarse dos veces, una por cada raíz de la ecuación indicial. El programa utiliza la subrutina “CoefPx” (ver figura 9) al que se le introducen dos parámetros: un polinomio y su grado, y devuelve una lista (contenida en la variable “T”) con los coeficientes del polinomio.

a. Para resolver la ecuación $2x^2 y'' - xy' + (1 + x^2)y = 0$ por el método de Frobenius, Tomamos $P = 2x^2$, $Q = -x$, $R = 1 + x^2$ y encontramos las raíces de la ecuación indicial mediante el

programa “Indicial” . Para esta ecuación obtenemos $r_1 = 1$ y $r_2 = \frac{1}{2}$.

Utilizamos ahora el programa “Frobe” con la primera de las raíces, $r_1 = 1$. El resultado se muestra en la figura 10.



Figura 9




Figura 10




Figura 11

Para la segunda raíz $r_2 = \frac{1}{2}$ el resultado se muestra en la figura 12.

La solución general puede escribirse en la forma:

$$y = Ax \left(1 + \frac{x^2}{10} + \frac{x^4}{360} + \dots \right) + B\sqrt{x} \left(1 - \frac{x^2}{6} + \frac{x^4}{168} + \dots \right)$$

Tarea 6. Modifique el programa “SerieP” para considerar ecuaciones no homogéneas, es decir ecuaciones de la forma $y'' + P(x)y' + Q(x)y = F(x)$.

- a. Con este nuevo programa, encuentre los primeros 6 términos de la solución en serie de potencia para la ecuación $y'' - 2x^2y' + 4xy = x^2 + 2x + 2$.

```

DelVar P, Q, N, T, E, E1, E2, x, sp, A, B
Input P, "P"
Input Q, "Q"
Input F, "F"
Input N, "N>3"
((P|x=0) * a1 + (Q|x=0) * a0 + F|x=0) / 2 => A
{a0, a1, A} => T
(P * diff(h(x), x) + Q * h(x) + F) => E
For 1 => k To N-3
simplify(diff(E, x)) => E
simplify(E) | diff(h(x), x) = a => E1
For 2 => j To k+1
simplify(E1) | diff(h(x), x, j) = j! * T[j+1] => E1
Next
simplify(E1) | h(x) = a0 | x=0 => E2
E2 / ((k+2)!) => B
augment(T, { B }) => T
Next
a0 => sp
For 2 => k To N
sp + T[k] * x^(k-1) => sp
Next
PrintNatural sp
DelVar P, Q, N, E, E1, E2

```

Solución.

El programa modificado que denominamos “SerNoHom” se muestra en el recuadro. Las modificaciones hechas están marcadas por cuadros.




Figura 12

Para aplicar este programa debemos escribir la ecuación en la forma:

$$y'' = P(x)y' + Q(x)y + F(x)$$

Por ejemplo para la ecuación $y'' - 2x^2y' + 4xy = x^2 + 2x + 2$ debemos escribir:

$y'' = \overbrace{2x^2}^P y' - \overbrace{4x}^Q y + \overbrace{x^2 + 2x + 2}^F$. El resultado de emplear el programa "SerNoHom" para esta ecuación con $N = 6$ se muestra en la figura 12